

2 x16 Parallel LCD (#603-00006)

General Information

The 2 X16 Parallel LCD is an 8 bit or 4 bit parallel interfaced LCD. This unit allows the user to display text, numerical data and custom created characters.

The LCD uses the HD44780 series LCD driver from Hitachi, or equivalent controller. The LCD is connected to a female 14-pin connector for easy interface with the BS2p24/40 Demo Board (#45187) and the Professional Development Board (#28138).




Though the device has the ribbon cable and 14-pin connector it may also be hooked up manually using the diagram on the next page.

Technical Specifications


Cable length: 6" (152 mm)
 Power requirements: 5.0 +VDC

Dimensions may vary




LCD Control from a BASIC Stamp

Parallax (www.parallax.com) publishes many circuits and examples to control LCDs. Most of these examples are available for download from our web site. These examples are featured in StampWorks, the Nuts and Volts of BASIC Stamps books, the free LCD Character Creator Software, and the BS2p Plus Pack. Example codes are listed below for the BASIC Stamp 1 and 2 modules.



To interface to the LCD in a 4-bit mode you will need set up the LCD in the following manner.
Stamp I/O pins



BASIC Stamp 1 code

```

' =====
' File..... Parallel LCD.bs1
' Purpose... Parallel LCD Display Demo
' Author... Parallax, Inc.
' E-mail... support@parallax.com
' {$STAMP BS1}
' {$PBASIC 1.0}
' ----[ Program Description ]-----
' This program demonstrates using a Hitachi-compatible Parallel LCD Display
' ----[ I/O Definitions ]-----
SYMBOL      E = 0           ' Enable Pin. 1 = Enabled
SYMBOL      RS = 3         ' Register Select. 0 = Instruction
SYMBOL      RW = 2         ' Read / Write control. 0 = Write
' ----[ Variables ]-----
SYMBOL      char = B3      ' Character sent to LCD
SYMBOL      temp = B4      ' Temp Variable
' ----[ EEPROM Data ]-----
EEPROM ("Hello, this is the LCD demo.")
' ----[ Initialization ]-----
Begin:
  LET PINS = 0             ' Clear The Output Lines
  LET DIRS = %11111000    ' One Input, 7 Outputs
  PAUSE 200               ' Wait 200 ms For LCD To Reset

```

```

LOW RW                                     ' Puts LCD Into Write Mode
Init LCD:
LET PINS = %00110000                       ' Wakes Up LCD
PULSOUT E,1                                ' Send Data Three Times
PAUSE 10
PULSOUT E,1
PAUSE 10
PULSOUT E,1
PAUSE 10
LET PINS = %00100000                       ' Function Set, 4-bit Operation
PULSOUT E,1
LET char = %00101000                       ' Set 4-bit Operation
GOSUB Wr LCD
LET char = 1                               ' Clears The LCD Screen
GOSUB Wr LCD
LET char = 6                               ' Set Cursor Direction
GOSUB Wr LCD
LET char = 14                              ' Sets Cursor To Underline
GOSUB Wr LCD
HIGH RS                                     ' Prepare To Send Characters
' -----[ Program Code ]-----
Main:
FOR B6 = 0 TO 27                            ' Read Data From EEPROM To Display
READ B6, char
IF B6 = 15 THEN Next Line
GOSUB Wr LCD
Out:
NEXT
END                                         ' End Of Code
' -----[ Subroutines ]-----
Wr LCD:
temp = char & %11110000                   ' Write The Character In B3 To LCD
                                           ' Logical AND Data Of High Byte Of
                                           ' I/O Pins
PINS = PINS & %00001000 | temp            ' Logical OR The Data Leaving RS
                                           ' Pin In An Unchanged State
PULSOUT E,1                                ' Clocks Out Data
PAUSE 10
temp = char & %00001111 * 16              ' Logical AND Data To Low Byte Of
                                           ' I/O Pins And Shifts To The Left
PINS = PINS & %00001000 | temp            ' Logical ORs The Data Leaving RS
                                           ' Pin In An Unchanged State
PULSOUT E,1                                ' Clocks Out Data
PAUSE 100
RETURN
Next line:
LOW RS
LET char = 128 + 64                        ' Places Cursor On Line 2
GOSUB Wr LCD
HIGH RS                                     ' Puts LCD Into Display Mode
READ B6, char
GOSUB Wr LCD
GOTO Out

```

BASIC Stamp 2, 2e and 2sx code

```

' =====
' File..... Parallel LCD 2X16.bs2
' Purpose... Parallel LCD Display Demo
' Author.... Parallax, Inc.
' E-mail.... support@parallax.com
' {$STAMP BS2}
' {$PBASIC 2.5}
' -----[ Program Description ]-----
' This program demonstrates using a Hitachi-compatible Parallel LCD Display
' This code works with the BS2, BS2e and BS2sx
' -----[ I/O Definitions ]-----
E          PIN      0          ' Enable Pin For LCD
RW         PIN      2          ' R/W Pin For LCD
RS         PIN      3          ' LCD Register Select
                                           ' 0 = Instruction, 1 = Text
' -----[ Variables ]-----
char       VAR      Byte      ' Character To Send To LCD
inst       VAR      char      ' Induction To Send To LCD
index      VAR      Word      ' Character Pointer
temp       VAR      Byte      ' Temp Variable
' -----[ EEPROM Data ]-----
DATA "Hello, this is the LCD demo."      ' Message To Send To LCD
' -----[ Initialization ]-----

```

```

Initialize:
  LOW RW                                ' Set LCD To Write Mode
  OUTS = %0000000000000000            ' Set All Output Low
  DIRS = %0000000011111111            ' Set I/O Direction
  GOSUB Init_Lcd                        ' Initialize The LCD Display
' -----[ Program Code ]-----
Main:
  FOR temp = 0 TO 27                    ' 28 Characters
    IF temp = 15 THEN                   ' Check For End Of Line
      GOSUB Next_Line                   ' Jump To Next Line
    ENDIF
    READ temp, char                      ' Read Next Character From EEPROM
    GOSUB Send_Text                     ' Send Character To LCD Display
  NEXT
END
' -----[ Subroutines ]-----
Init_Lcd:
  PAUSE 200
  OUTS = %00110000                      ' Reset The LCD
  PULSOUT E,1                           ' Send Command Three Times
  PAUSE 10
  PULSOUT E,1
  PAUSE 10
  PULSOUT E,1
  PAUSE 10
  OUTS = %00100000                      ' Set To 4-bit Operation
  PULSOUT E,1
  Inst = %00101000                      ' Function Set (2-Line Mode)
  GOSUB Send_Inst
  Inst = %00001110                      ' Turn On Cursor
  GOSUB Send_Inst
  Inst = %00000110                      ' Set Auto-Increment
  GOSUB Send_Inst
  Inst = %00000001                      ' Clears LCD
  GOSUB Send_Inst
  Inst = 14                             ' Set Cursor To Underline
  GOSUB Send_Inst
  RETURN
Send_Inst:
  LOW RS                                ' Set Instruction Mode
  OUTB = Inst.HIGHNIB                   ' Send High Nibble
  PULSOUT E,1
  OUTB = Inst.LOWNIB                    ' Send Low Nibble
  PULSOUT E,1
  HIGH RS                               ' Set LCD Back To Text Mode
  RETURN
Send_Text:
  OUTB = Char.HIGHNIB                   ' Send High Nibble
  PULSOUT E,1
  OUTB = Char.LOWNIB                    ' Send Low Nibble
  PULSOUT E,1
  PAUSE 100
  RETURN
Next_Line:
  Inst = 128+64                         ' Move Cursor To Line 2
  GOSUB Send_Inst
  RETURN

```

BASIC Stamp 2p24, 2p40 2pe and 2px24 code

```

' =====
' File..... Parallel LCD 2X16.bsp
' Purpose... Parallel LCD Display Demo
' Author.... Parallax, Inc.
' E-mail.... support@parallax.com
' {$STAMP BS2p}
' {$PBASIC 2.5}
' -----[ Program Description ]-----
' This program demonstrates using a Hitachi-compatible Parallel LCD Display
' This code works with the BS2p24, BS2p40, BS2pe and BS2px24
' -----[ I/O Definitions ]-----
Lcd      PIN      0          ' LCD Enable Pin
' -----[ Variables ]-----
temp     VAR      Byte
char     VAR      Byte
' -----[ Initialization ]-----
Init_Lcd:

```

```

PAUSE 1000
FOR temp = 0 TO 2
  LCDCMD Lcd, 48          ' Reset LCD (Send 3 times)
  PAUSE 5                ' Delay Require By LCD Specs
NEXT
LCDCMD Lcd, 32          ' Set 4-bit Mode
LCDCMD Lcd, 40          ' Set 2-line Mode
LCDCMD Lcd, 12         ' Turn On Display With No Cursor
LCDCMD Lcd, 6          ' Set To Auto-Increment Cursor
LCDCMD Lcd, 1          ' Clear Display
' -----[ Main Routine ]-----
Main:
DO
  LCDOUT Lcd, 1, ["Hello, this is"] ' Clear LCD & Print Line 1 Text
  LCDOUT Lcd, 192, ["the LCD demo."] ' Move To Line 2 & Print Text
  PAUSE 3000                       ' Wait A Few Seconds
  LCDCMD Lcd, 1                     ' Clear LCD
  PAUSE 500                         ' Wait 1/2 Second
LOOP                                 ' Do It Again
END

```

Компания «Океан Электроники» предлагает заключение долгосрочных отношений при поставках импортных электронных компонентов на взаимовыгодных условиях!

Наши преимущества:

- Поставка оригинальных импортных электронных компонентов напрямую с производств Америки, Европы и Азии, а так же с крупнейших складов мира;
- Широкая линейка поставок активных и пассивных импортных электронных компонентов (более 30 млн. наименований);
- Поставка сложных, дефицитных, либо снятых с производства позиций;
- Оперативные сроки поставки под заказ (от 5 рабочих дней);
- Экспресс доставка в любую точку России;
- Помощь Конструкторского Отдела и консультации квалифицированных инженеров;
- Техническая поддержка проекта, помощь в подборе аналогов, поставка прототипов;
- Поставка электронных компонентов под контролем ВП;
- Система менеджмента качества сертифицирована по Международному стандарту ISO 9001;
- При необходимости вся продукция военного и аэрокосмического назначения проходит испытания и сертификацию в лаборатории (по согласованию с заказчиком);
- Поставка специализированных компонентов военного и аэрокосмического уровня качества (Xilinx, Altera, Analog Devices, Intersil, Interpoint, Microsemi, Actel, Aeroflex, Peregrine, VPT, Syfer, Eurofarad, Texas Instruments, MS Kennedy, Miteq, Cobham, E2V, MA-COM, Hittite, Mini-Circuits, General Dynamics и др.);

Компания «Океан Электроники» является официальным дистрибьютором и эксклюзивным представителем в России одного из крупнейших производителей разъемов военного и аэрокосмического назначения «JONHON», а так же официальным дистрибьютором и эксклюзивным представителем в России производителя высокотехнологичных и надежных решений для передачи СВЧ сигналов «FORSTAR».



JONHON

«JONHON» (основан в 1970 г.)

Разъемы специального, военного и аэрокосмического назначения:

(Применяются в военной, авиационной, аэрокосмической, морской, железнодорожной, горно- и нефтедобывающей отраслях промышленности)

«FORSTAR» (основан в 1998 г.)

ВЧ соединители, коаксиальные кабели,
кабельные сборки и микроволновые компоненты:

(Применяются в телекоммуникациях гражданского и специального назначения, в средствах связи, РЛС, а так же военной, авиационной и аэрокосмической отраслях промышленности).



Телефон: 8 (812) 309-75-97 (многоканальный)

Факс: 8 (812) 320-03-32

Электронная почта: ocean@oceanchips.ru

Web: <http://oceanchips.ru/>

Адрес: 198099, г. Санкт-Петербург, ул. Калинина, д. 2, корп. 4, лит. А